
Potpourri of Pupil Masks

PSF for Single Prolate Spheroidal Pupil

Fig. 5.— Left The single prolate spheroidal wave function shaped pupil aperture (Slepian

1965) inscribed in a circular aperture of unit area. Right The corresponding PSF plotted on

a logarithmic scale with black areas 10−10 below brightest. This mask has a single-exposure

normalized discovery integration time of 4.6 with a small discovery space at the inner working

distance (IWD).

Pupil Apodization

Let f (x, y) denote the transmissivity (i.e., apodization) at location (x, y) on the surface of
a filter placed in the pupil of a telescope.

The electromagnetic field in the image plane of such a telescope associated with an on-axis
point source (i.e., a star) is proportional to the Fourier transform of the filter function f .

Choosing units in such a way that the telescope’s opening has a diameter of one, the Fourier
transform can be written as

f̂ (ξ, η) =

∫ 1/2

−1/2

∫ 1/2

−1/2

e2πi(xξ+yη)f (x, y)dxdy. (1)

The intensity of the light in the image is proportional to the magnitude squared of the
electromagnetic field.

Assuming that the underlying telescope has a circular opening of diameter one, we impose
the following constraint on the function f :

f (x, y) = 0 for x2 + y2 > (1/2)2.



Optimized Apodizations

Multiple competing goals: (i) maximize throughput vs. (ii) minimize light in dark zone D of
the image plane vs. (iii) maximize size of D.

If too much light lands in the dark zone, the telescope will fail to detect the planets it is
designed to find. Hence, objective (ii) is usually formulated as a constraint.

This leads to the following optimization problem:

maximize

∫∫
f (x, y)dxdy

(
= f̂ (0, 0)

)
subject to

∣∣∣f̂ (ξ, η)
∣∣∣2≤ ε f̂ (0, 0)2, (ξ, η) ∈ D,

f (x, y) = 0, x2 + y2 > (1/2)2,
0 ≤ f (x, y) ≤ 1, for all x, y.

(2)

Here, ε is a small positive constant representing the maximum level of brightness of the
starlight in the dark zone.

Without imposing further symmetry constraints on the function f , the Fourier transform f̂
is complex valued.

This optimization problem has a linear objective function and both linear constraints and
convex quadratic inequality constraints. Hence, a discretized version can be solved (to a
global optimum).



Exploiting Symmetry

Assuming that the filter can be symmetric with respect to reflection about both axes (note:
sometimes not possible), the Fourier transform can be written as

f̂ (ξ, η) = 4

∫ 1/2

0

∫ 1/2

0

cos(2πxξ) cos(2πyη)f (x, y)dxdy.

In this case, the Fourier transform is real and so the convex quadratic inequality constraint
in (2) can be replaced with a pair of inequalities,

−
√
ε f̂ (0, 0) ≤ f̂ (ξ, η) ≤

√
ε f̂ (0, 0),

making the problem an infinite dimensional linear programming problem.



Discretization

Consider a two-dimensional Fourier transform

f̂ (ξ, η) = 4

∫ 1/2

0

∫ 1/2

0

cos(2πxξ) cos(2πyη)f (x, y)dxdy.

Its discrete approximation can be computed as

f̂j1,j2 = 4
n∑

k2=1

n∑
k1=1

cos(2πxk1ξj1) cos(2πyk2ηj2)fk1,k2∆x∆y, 1 ≤ j1, j2 ≤ m, (3)

where
xk = (k − 1/2)∆x, 1 ≤ k ≤ n,

yk = (k − 1/2)∆y, 1 ≤ k ≤ n,

ξj = (j − 1/2)∆ξ, 1 ≤ j ≤ m,

ηj = (j − 1/2)∆η, 1 ≤ j ≤ m,

fk1,k2 = f (xk1, yk2), 1 ≤ k1, k2 ≤ n

f̂j1,j2 ≈ f̂ (ξj1, ηj2), 1 ≤ j1, j2 ≤ m.

Complexity: m2n2.



A Clever(?) Idea

Performing the calculation in the obvious (brute force) way requires m2n2 complex additions
and a similar number of multiplies. However, we can factor the double sum into two nested
single sums. Introducing new variables that represent the inner sum, we get:

gj1,k2 = 2
n∑

k1=1

cos(2πxk1ξj1)fk1,k2∆x, 1 ≤ j1 ≤ m, 1 ≤ k2 ≤ n,

f̂j1,j2 = 2
n∑

k2=1

cos(2πyk2ηj2)gj1,k2∆y, 1 ≤ j1, j2 ≤ m,

Formulated this way, the calculation requires only mn2+m2n additions (and a similar number
of multiplications).



Brute Force vs Clever Approach

On the following page we show two ampl model formulations of this problem.

On the left is the version expressed in the straightforward one-step manner.

On the right is the ampl model for the same problem but with the Fourier transform
expressed as a pair of transforms—the so-called two-step process.

The dark zone D is a pair of sectors of an annulus with inner radius 4 and outer radius 20.

Except for the resolution, the two models produce essentially the same result.



Two ampl Models

param pi := 4*atan(1);

param rho0 := 4;
param rho1 := 20;
param m := 35; # discretization parameter
param n := 150; # discretization parameter
param dx := 1/(2*n);
param dy := dx;

set Xs := setof {j in 0.5..n-0.5 by 1} j/(2*n);
set Ys := setof {j in 0.5..n-0.5 by 1} j/(2*n);
set Pupil :=

setof {x in Xs, y in Ys: x^2+y^2<0.25} (x,y);
set Xis := setof {j in 0..m} j*rho1/m;
set Etas := setof {j in 0..m} j*rho1/m;
set DarkHole := setof {xi in Xis, eta in Etas:

xi^2+eta^2>=rho0^2 &&
xi^2+eta^2<=rho1^2 &&
eta <= xi } (xi,eta);

var f {(x,y) in Pupil} >= 0, <= 1;

var fhat {xi in Xis, eta in Etas};

maximize area: sum {(x,y) in Pupil} f[x,y]*dx*dy;

subject to fhat_def {xi in Xis, eta in Etas}:
fhat[xi,eta] = 4*sum {(x,y) in Pupil}

f[x,y]*cos(2*pi*x*xi)
*cos(2*pi*y*eta)*dx*dy;

subject to sidelobe_pos {(xi,eta) in DarkHole}:
fhat[xi,eta] <= 10^(-5)*fhat[0,0];

subject to sidelobe_neg {(xi,eta) in DarkHole}:
-10^(-5)*fhat[0,0] <= fhat[xi,eta];

solve;

param pi := 4*atan(1);

param rho0 := 4;
param rho1 := 20;
param m := 35; # discretization parameter
param n := 1000; # discretization parameter
param dx := 1/(2*n);
param dy := dx;

set Xs := setof {j in 0.5..n-0.5 by 1} j/(2*n);
set Ys := setof {j in 0.5..n-0.5 by 1} j/(2*n);
set Pupil :=

setof {x in Xs, y in Ys: x^2+y^2 < 0.25} (x,y);
set Xis := setof {j in 0..m} j*rho1/m;
set Etas := setof {j in 0..m} j*rho1/m;
set DarkHole := setof {xi in Xis, eta in Etas:

xi^2+eta^2>=rho0^2 &&
xi^2+eta^2<=rho1^2 &&
eta <= xi } (xi,eta);

var f {(x,y) in Pupil} >= 0, <= 1;
var g {xi in Xis, y in Ys};
var fhat {xi in Xis, eta in Etas};

maximize area: sum {(x,y) in Pupil} f[x,y]*dx*dy;

subject to g_def {xi in Xis, y in Ys}:
g[xi,y] = 2*sum {x in Xs: (x,y) in Pupil}

f[x,y]*cos(2*pi*x*xi)*dx;
subject to fhat_def {xi in Xis, eta in Etas}:

fhat[xi,eta] = 2*sum {y in Ys}
g[xi,y]*cos(2*pi*y*eta)*dy;

subject to sidelobe_pos {(xi,eta) in DarkHole}:
fhat[xi,eta] <= 10^(-5)*fhat[0,0];

subject to sidelobe_neg {(xi,eta) in DarkHole}:
-10^(-5)*fhat[0,0] <= fhat[xi,eta];

solve;



The Optimal Solution

Left. The optimal apodization found by either of the models shown on previous slide.

Center. Plot of the star’s image (using a linear stretch).

Right. Logarithmic plot of the star’s image (black = 10−10).

Notes:

• The “apodization” turns out to be purely opaque and transparent (i.e., a mask).

• The mask has “islands” and therefore must be laid on glass.



Close Up

Brute force with n = 150

Two-step with n = 1000

Summary Problem Stats

Comparison between a few sizes of the one and two step models.

Problem-specific stats.

Model n m constraints variables nonzeros arith. ops.
One step 150 35 976 17,672 17,247,872 17,196,541,336
One step 250 35 * * * *
Two step 150 35 7,672 24,368 839,240 3,972,909,664
Two step 500 35 20,272 215,660 7,738,352 11,854,305,444
Two step 1000 35 38,272 822,715 29,610,332 23,532,807,719

Hardware/Solution-specific performance comparison data.

Model n m iterations primal objective dual objective cpu time (sec)
One step 150 35 54 0.05374227247 0.05374228041 1380
One step 250 35 * * * *
Two step 150 35 185 0.05374233071 0.05374236091 1064
Two step 500 35 187 0.05395622255 0.05395623990 4922
Two step 1000 35 444 0.05394366337 0.05394369256 26060



Concentric Ring Masks



Circular Unobstructed Pupil



Circular Unobstructed Pupil



Circular Unobstructed Pupil

Matrix Notation

As Fourier transforms are linear operators it is instructive to express them in matrix/vector
notation. To this end, let

F := [fk1,k2], G := [gj1,k2], F̂ := [f̂j1,j2], and K := [κj1,k2],

where K denotes the m× n Fourier kernel matrix whose elements are

κj1,k2 = 2 cos(2πxk1ξj1)∆x.

Then, the two-dimensional Fourier transform F̂ can be written simply as

F̂ = KFKT

and the computation of the transform in two steps is just the statement that the two matrix
multiplications can, and should, be done separately:

G = KF

F̂ = GKT .



matlab

Brute force method:

for j1 in 1:m
for j2 in 1:m

fhat(j1,j2) = 0;

for k1 in 1:n

for k2 in 1:n

fhat(j1,j2) = fhat(j1,j2) + ...

4 * cos(2*pi*x(k1)*xi(j1)) * ...

cos(2*pi*y(k2)*eta(j2)) * ...

f(k1,k2)*dx*dy;

end

end

end

end

Smart method:

K = 2 * cos(2*pi*xi*x’)*dx;

fhat = K*f*K’;



Linear Programming Paradigm

Linear programming software solves problems in this form:

maximize cTx
subject to Ax= b,

x ≥ 0,

where b and c are given vectors and A is a given matrix. To be nontrivial, we need to assume
that A has more columns than rows.

And, of course, x is a vector.

ampl converts problems from their “natural” formulation to this paradigm and then invokes
a solver.



Converting to the LP Paradigm

Let fj, gj, and f̂j denote the column vectors of matrices F , G, and F̂ :

F = [f1 f2 · · · fn] , G = [g1 g2 · · · gn] , F̂ =
[
f̂1 f̂2 · · · f̂m

]
.

We can list the elements of F , G and F̂ in column vectors:

vec(F ) =


f1

f2
...
fn

 , vec(G) =


g1

g2
...
gn

 , vec(F̂ ) =


f̂1

f̂2
...

f̂m

 .
It is straightforward to check that

vec(G) =


K

K
. . .

K

 vec(F )

and that

vec(F̂ ) =


κ1,1I κ1,2I · · · κ1,nI
κ2,1I κ2,2I · · · κ2,nI

... ... . . . ...
κm,1I κm,2I · · · κm,nI

 vec(G).



The Constraint Matrix

One-Step Method

Ax = b

m


κ1,1K κ1,2K · · · κ1,nK −I
κ2,1K κ2,2K · · · κ2,nK −I

... ... . . . ... . . .
κm,1K κm,2K · · · κm,nK −I

... ...





f1

f2
...
fn

f̂1

f̂2
...

f̂m


=


0
0
...
0
...



The big left block is a dense m2 × n2 matrix.



The Constraint Matrix: Two-Step Method

Ax = b

m



K −I
K −I

. . . . . .
K −I

κ1,1I κ1,2I · · · κ1,nI −I
κ2,1I κ2,2I · · · κ2,nI −I

... ... . . . ... . . .
κm,1I κm,2I · · · κm,nI −I

... ... ...





f1

f2
...
fn

g1

g2
...
gn

f̂1

f̂2
...

f̂m



=



0
0
...
0
0
0
...
0
...



The big upper-left block is a sparse block-diagonal mn× n2 matrix with m2n nonzeros.
The middle block is an m×n matrix of sub-blocks which are each m×m diagonal matrices.
Hence, it is very sparse, containing only m2n nonzeros.



Fast Fourier Transform (in 1D)—It’s the same idea!

One-dimensional Fourier transform:

f̂ (ξ) =

∫ 1/2

−1/2

e2πixξf (x)dx.

Discrete approximation:

f̂j =
n∑

k=−n

e2πik∆xj∆ξfk∆x, −m ≤ j ≤ m.

Let
N = 2n + 1 and M = 2m + 1.

Suppose that N and M can be factored:

N = N0N1 and M = M0M1.

Since M and N are odd, the factors must be odd too:

N0 = 2n0 + 1, N1 = 2n1 + 1, M0 = 2m0 + 1, and M1 = 2m1 + 1.



Fast Fourier Transform (FFT)

If we now decompose our sequencing indices k and j into

k = N0k1 + k0 and j = M0j1 + j0,

we get

f̂j0,j1 =
n0∑

k0=−n0

n1∑
k1=−n1

e2πiN0k1∆xM0j1∆ξ e2πiN0k1∆xj0∆ξ e2πik0∆x(M0j1+j0)∆ξ fk0,k1∆x.

We want the first exponential factor to evaluate to one. To make that happen, we assume
that N0M0∆x∆ξ is an integer. With that first exponential factor out of the way, we can
again write down a two-step algorithm

gj0,k0 =
n1∑

k1=−n1

e2πiN0k1∆xj0∆ξ fk0,k1∆x,
−m0 ≤ j0 ≤ m0,
−n0 ≤ k0 ≤ n0,

f̂j0,j1 =
n0∑

k0=−n0

e2πik0∆x(M0j1+j0)∆ξgj0,k0,
−m0 ≤ j0 ≤ m0

−m1 ≤ j1 ≤ m1.



Complexity

The number of multiply/adds required for this two-step algorithm is

NM0 + MN0 = MN

(
1

M1

+
1

N1

)
.

If M ≈ N and M1 ≈ N1 ≈
√
N , the complexity simplifies to

2N
√
N.

Compared to the one-step algorithm, which takes N 2 multiply/adds, this two-step algorithm
gives an improvement of a factor of

√
N/2. Also, if M is much smaller than N , we get

further improvement over the full N ×N case.

Of course, if M0,M1, N0, and N1 can be further factored, then this two-step algorithm can
be extended recursively.

For the FFT, M and N are chosen to be a power of 3, or more commonly 2. In this case,
the recursively applied algorithm is an N log2N algorithm.

